

NAG Library Routine Document

F03BAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F03BAF computes the determinant of a real n by n matrix A . F07ADF (DGETRF) must be called first to supply the matrix A in factorized form.

2 Specification

```
SUBROUTINE F03BAF (N, A, LDA, IPIV, D, ID, IFAIL)
```

```
INTEGER          N, LDA, IPIV(N), ID, IFAIL
```

```
REAL (KIND=nag_wp) A(LDA,*), D
```

3 Description

F03BAF computes the determinant of a real n by n matrix A that has been factorized by a call to F07ADF (DGETRF). The determinant of A is the product of the diagonal elements of U with the correct sign determined by the row interchanges.

4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer-Verlag

5 Parameters

- | | | |
|----|---|---------------|
| 1: | N – INTEGER | <i>Input</i> |
| | <i>On entry:</i> n , the order of the matrix A . | |
| | <i>Constraint:</i> $N > 0$. | |
| 2: | A(LDA,*) – REAL (KIND=nag_wp) array | <i>Input</i> |
| | Note: the second dimension of the array A must be at least N . | |
| | <i>On entry:</i> the n by n matrix A in factorized form as returned by F07ADF (DGETRF). | |
| 3: | LDA – INTEGER | <i>Input</i> |
| | <i>On entry:</i> the first dimension of the array A as declared in the (sub)program from which F03BAF is called. | |
| | <i>Constraint:</i> $LDA \geq N$. | |
| 4: | IPIV(N) – INTEGER array | <i>Input</i> |
| | <i>On entry:</i> the row interchanges used to factorize matrix A as returned by F07ADF (DGETRF). | |
| 5: | D – REAL (KIND=nag_wp) | <i>Output</i> |
| 6: | ID – INTEGER | <i>Output</i> |
| | <i>On exit:</i> the determinant of A is given by $D \times 2.0^{ID}$. It is given in this form to avoid overflow or underflow. | |

7: IFAIL – INTEGER

Input/Output

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $N < 1$.

IFAIL = 3

On entry, $LDA < N$.

IFAIL = 4

The matrix A is approximately singular.

7 Accuracy

The accuracy of the determinant depends on the conditioning of the original matrix. For a detailed error analysis, see page 107 of Wilkinson and Reinsch (1971).

8 Further Comments

The time taken by F03BAF is approximately proportional to n .

9 Example

This example computes the LU factorization with partial pivoting, and calculates the determinant, of the real matrix

$$\begin{pmatrix} 33 & 16 & 72 \\ -24 & -10 & -57 \\ -8 & -4 & -17 \end{pmatrix}.$$

9.1 Program Text

```

Program f03baf

!      F03BAF Example Program Text

!      Mark 24 Release. NAG Copyright 2012.

!      .. Use Statements ..
      Use nag_library, Only: dgetrf, f03baf, nag_wp, x04caf
!      .. Implicit None Statement ..
      Implicit None

```

```

! .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
! .. Local Scalars ..
Real (Kind=nag_wp)         :: d
Integer                    :: i, id, ifail, info, lda, n
! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: a(:, :)
Integer, Allocatable        :: ipiv(:)
! .. Executable Statements ..
Write (nout,*) 'F03BAF Example Program Results'

! Skip heading in data file
Read (nin,*)

Read (nin,*) n
lda = n
Allocate (a(lda,n),ipiv(n))

Read (nin,*)(a(i,1:n),i=1,n)

! LU Factorize A
! The NAG name equivalent of dgetrf is f07adf
Call dgetrf(n,n,a,lda,ipiv,info)

Write (nout,*)
Flush (nout)
ifail = 0
Call x04caf('G','N',n,n,a,lda,'Array A after factorization',ifail)

Write (nout,*)
Write (nout,*) 'Pivots'
Write (nout,99999) ipiv(1:n)
Write (nout,*)
Flush (nout)

ifail = 0
Call f03baf(n,a,lda,ipiv,d,id,ifail)

Write (nout,99998) d, id
Write (nout,*)
Write (nout,99997) d*2.0_nag_wp**id

99999 Format (1X,8(1X,I13))
99998 Format (1X,'D = ',F13.5,' ID = ',I12)
99997 Format (1X,'Value of determinant = ',E13.5)
End Program f03baf

```

9.2 Program Data

F03BAF Example Program Data

```

3           : N
 33   16   72
-24  -10  -57
  -8   -4  -17           : A

```

9.3 Program Results

F03BAF Example Program Results

Array A after factorization

| | 1 | 2 | 3 |
|---|---------|---------|---------|
| 1 | 33.0000 | 16.0000 | 72.0000 |
| 2 | -0.7273 | 1.6364 | -4.6364 |
| 3 | -0.2424 | -0.0741 | 0.1111 |

Pivots

| | | | |
|------------------------|-------------|------|---|
| | 1 | 2 | 3 |
| D = | 0.37500 | ID = | 4 |
| Value of determinant = | 0.60000E+01 | | |
